[RADIOCARBON, VOL 28, NO. 2B, 1986, P 1022–1030]

A COMPUTER PROGRAM FOR RADIOCARBON AGE CALIBRATION

MINZE STUIVER and PAULA J REIMER

Department of Geological Sciences and Quaternary Research Center
University of Washington, Seattle, Washington 98195

The calibration curves and tables given in this issue of RADIOCARBON form a data base ideally suited for a computerized operation. The program listed below converts a radiocarbon age and its age error o_s (one standard deviation) into calibrated ages (intercepts with the calibration curve), and ranges of calibrated ages that correspond to the age error. The standard deviation o_c in the calibration curve is taken into account using $o_{total} = \sqrt{o_s^2 + o_c^2}$ (see Stuiver and Pearson, this issue, for details).

The program transforms radiocarbon ages into cal AD/BC(cal BP) ages. Probabilities within the cal age ranges are not included, this feature will be incorporated at later stage. The FORTRAN program and calibration data can be obtained for the cost of materials and shipping (US $5, prepaid and payable to the Quaternary Research Center) from the Quaternary Isotope Laboratory on a DS/DD floppy diskette. The calibration data were assembled from this calibration issue, and from the tabulations of Linick, Suess, and and Becker (Radiocarbon, 27, 20-32, 1985).

The current commitment of the Quaternary Isotope Laboratory is to supply the 1986 version of the program. We do not yet pledge continuous updating, but will make an attempt if time and budget permit. The program is IBM PC-XT compatible; users are responsible for adaptation to non-compatible systems. A visual display (although not given here) is part of the floppy disk version. Future use of the program will surely lead to modifications and we welcome suggestions.

```
C
C Radiocarbon Calibration Program CALIB
C
C     The program converts radiocarbon ages to calibrated ages as
C  would be done if one manually plotted the calibration curve data*
C  on an X-Y axis and drew a line through the Y-axis corresponding to
C  the radiocarbon age.  Vertical lines drawn through these intercepts
C  to the X-axis, with linear interpolation between data points, give
C  the cal AD/BC ages.  Cal BP ages are calculated from 1950 so that
C  cal BP = 1950 - cal AD and cal BP = 1949 + cal BC.  The one year
C  difference in converting BC dates is caused by the absence of the
C  zero year in the AD/BC chronology.
C     To convert the standard error in the radiocarbon age into a range
C  of cal AD/BC (BP) ages the user must first determine whether to use
C  1) the laboratory quoted error or 2) increase the quoted error by a
C  known "error multiplier"  (Stuiver and Pearson, 1986, Radiocarbon,
C  28, 805-838.)  With the sample sigma entered,the program calculates
C  the total sigma for non-marine samples as:
C
C    1 Sigma = SQRT((sample sigma)^2 + (calibration curve sigma)^2)
C    2 Sigma = SQRT((2*sample sigma)^2 + (calibration curve sigma)^2)
C
C  (Stuiver, 1982, Radiocarbon, 24, 1-26). The calibration curve sigma
C  is the average of the standard deviaton of the 2 data points closest
C  to each intercept of the radiocarbon age Y.  Vertical lines drawn to
C  the X-axis through the intercepts of Y + 1 Sigma and Y - 1 Sigma with
C  the calibration curve give the ranges of cal AD/BC ages for 1 Sigma.
C  Likewise intercepts of Y + 2 Sigma and Y - 2 Sigma give the 2 Sigma
C  ranges.  For ranges and sample sigmas greater than 100 years the
C  ranges are rounded to the nearest ten years.  Ranges that overlap or
C  are closer together than one year, or ten if rounded, are reported as
C  one age range.
C     Marine samples are treated similarly except that the user must
C  determine the Delta R and the uncertainty in Delta R to use for
C  each sample based on its collection location (Stuiver ,Pearson, and
C  Braziunas, 1986, Radiocarbon, 28, 2B...)  The marine total sigma is
C  taken as:
C
C    1 Sigma = SQRT((sample sigma)^2 + (Delta R sigma)^2)
C    2 Sigma = SQRT((2*sample sigma)^2 + (Delta R sigma)^2).
C
C     Three datasets are provided. The twenty year atmospheric record
C  (2) is recommended for most non-marine samples although a ten year
C  record (1) is given for more detailed comparisons of younger samples.
C  The 20 year marine record (3) should be used with all marine samples.
C
C *Input from files:
C
C     1. ATM10.14C
C        10 yr atmospheric record to 2490 cal BC (circa 4200 14-C BP)
C     2. ATM20.14C
C        20 yr atmospheric record to 7210 cal BC (circa 8200 14-C BP)
C     Format(1. and 2.):  Year, Radiocarbon age, Sigma age
C                         (5X,F9.1,5X,I5,5X,I2)
C     3. MARINE.14C
C        20 yr marine model record to 7190 cal BC (circa 8585 14-C BP)
C          Format:  Year, Radiocarbon age
C                   (5X,F9.1,5X,I5)
C
C Output :
C
C     1. to printer LPT1 if desired
C
C     2. OUTFIL.14C   for listing, rename to save
C
C     3. PLTFIL.14C   for plotting
C          Format: sample id, # of intercepts, calibrated ages
C                  1 sigma value, # of ranges, ranges,
C                  2 sigma value, # of ranges, ranges
C                  (1X,A12,I2,n(F10.1,2X))
C                  2(1X,F8.1,I2,r(F10.1,2X))
C                      where n=repeat spec.= # of intercepts
C                      where r=repeat spec.= # of ranges
C
C Subroutines:
C
C   INRCP  to find  the intercept of a radiocarbon age with the
C          calibration curve
C   Calling sequence:  CALL INRCP(V,NPTS,INTPT,NINP)
C     where:
C           V = Y value (Radiocarbon age) to intercept curve
C           NPTS = # of data points  (current dimension = 1000)
C           INTPT = array of intercepting points (max = 40)
```

```
C           NINP = # of intercepting points
C
C   ABWRT   writes calibrated age to unit LO
C
C   BPWRT   writes age BP to unit LO
C
C   RWRT    write age ranges to unit LO
C
C Revision date: 7/18/86
C PJ Reimer
C Quaternary Isotope Laboratory
C Quaternary Research Center AK-60
C University of Washington
C Seattle, WA 98195
C
$STORAGE:2
      PROGRAM CALIB
C
      COMMON X(1000),Y(1000),S(1000)
      COMMON /WRRNG/ RANGE,SIGMA1,NRANG,NPTS,JAD
      COMMON /WRINT/ ABINT,BPINT,ENLNE
C
      INTEGER*2 AGE,ABINT,BPINT,INTX(40,2)
      INTEGER*2 IRANGE(20,2)
      INTEGER*2 LU(3),MINTX(40,2),MENT,NENT,NINP,NINTX(40,2)
      INTEGER*2 RANK(80),SIGAGE,TREF
C
      REAL*4 DELTAR,EINT(40)
      REAL*4 INTPT(40),MINT(40),RAGE,RANGE(20,2)
      REAL*4 REFDAT(4,2),SAMSIG,SIG1,SIG2,SIGMA1,TEMPR(40),UNCR
      REAL*4 YMAX
C
      CHARACTER COMMA*1,CHSIG*1
      CHARACTER CHPM*1,CHSQD*1,FMT*100,IAD(3)*2,ICL*4
      CHARACTER ID*2,IDSAM*12,JAD*2,LP*1,MREF(5)*1,NAME*10,NAMOUT*10
      CHARACTER NAMPLT*10,NREF(6)*1,SREF(6)*21,SREF2(6)*21,SREF1*21
      CHARACTER REF1*63,REF2(5)*63,REF3(2)*63,REFAL(9)*63
      CHARACTER SREF3(2)*21,STR1*15,STR2*16
C
      LOGICAL SKIP,ENLNE
C
      EQUIVALENCE (ID,IDSAM)
C
      DATA COMMA/','/,FMT/' '/
      DATA LU/0,200,6/
      DATA NAMOUT/'OUTFIL.14C'/,NAMPLT/'PLTFIL.14C'/
      DATA NAME/'CAL20.14C'/,IAD/'AD','BP','AD'/,ICL/'cal '/
C
C Reference for 10 yr atmospheric record
C
      DATA REF1/'Stuiver,M and Becker,B, 1986, Radiocarbon, 28, 2B....'/
      DATA SREF1/'(Stuiver and Becker) '/
C
C References for 20 yr atmospheric record
C
      DATA REF2/
     &'Stuiver, M and Pearson, GW, 1986, Radiocarbon, 28, 805-838.',
     &'Pearson, GW and Stuiver, M, 1986, Radiocarbon, 28, 839-632.   ',
     &'Pearson, GW, Pilcher, JR, Baille, MG, Corbett, DM and Qua, F,  ',
     &'1986, Radiocarbon, 28, 2B.....                                 ',
     &'Bidecadal weighted average of data from:                      '/
      DATA REFAL/'Linick, TW, Suess, HE and Becker, B, (LSB) 1985, ',
     &'Radiocarbon, 27, 20-32. [for the interval 5219-7199 BC     ',
     &'Stuiver, M, Kromer, B, Becker, B, and Ferguson, CW, (SKBF)   ',
     &'1986, Radiocarbon, 28, 2B....',
     &'Kromer, B, Rhein, M, Bruns, M, Schoh-Fischer, H, Munnich, KO,',
     &'Stuiver, M, and Becker, B, (KRBSMSB) 1986, Radiocarbon, 28,',
     &'2B..... [for the interval 5229 -7207 BC]',
     &'Linick, TW, Long, A, Damon, PE and Ferguson, CW, (LLDF)  1986.',
     &'Radiocarbon, 28, 2B.'/
C
      DATA SREF2/'(Stuiver and Pearson)','(Pearson and Stuiver)',
     &'(Pearson et al. 1986)','(20 yr. average of     ',
     &' LSB,SKBF,KRBSMSB,',' and LLDF)'/
C
      DATA REFDAT/1955.,-500.,-2500.,-5230.,-500.,-2490.,-5210.,-7210./
C
C Reference for 20 yr marine record
C
      DATA REF3/'Stuiver, M, Pearson, GW, and Braziunas, T, 1986.',
     &'Radiocarbon, 28, 2B.'/
      DATA SREF3/'(Stuiver, Pearson and',' Braziunas)        '/
C
      DATA STR1/'Calibrated age:'/,STR2/'Calibrated ages:'/
C
C  define character Plus and minus, sigma, and squared and formfeed
C
      CHPM = CHAR(241)
      CHSIG = CHAR(229)
      CHSQD = CHAR(253)
C
C Open files for text output and plotting
C
      OPEN(6,FILE='LPT1')
      OPEN(200,FILE=NAMOUT,STATUS='NEW')
      OPEN(300,FILE=NAMPLT,STATUS='NEW')
      DO 20  IWRITE=1,3
         LO=LU(IWRITE)
         IF(IWRITE.GT.2) THEN
            WRITE(*,25)
25          FORMAT(1X,'Output to Printer?  Y(es) or N(o) ')
            READ(*,'(A)') LP
            IF((LP.NE.'Y').AND.(LP.NE.'y')) THEN
               LEND = 2
               GOTO 20
            ELSE
               LEND = 3
            ENDIF
         ENDIF
         WRITE(LO,'(26X,A)') 'UNIVERSITY OF WASHINGTON'
         WRITE(LO,'(27X,A)') 'QUATERNARY ISOTOPE LAB'
         WRITE(LO,'(23X,A)') 'RADIOCARBON CALIBRATION PROGRAM 1986'
         WRITE(LO,*)
20     CONTINUE
       WRITE(*,30)
30     FORMAT(/,1X,'Select calibration curve dataset.',/,1X,
     &'1. 10 yr atmospheric record to 2490 cal BC (circa 4200 14-C BP)',
     &/,1X,
     &'2. 20 yr atmospheric record to 7210 cal BC (circa 8200 14-C BP)',
     &/,1X,
     &'3. 20 yr marine model to 7190 cal BC (circa 8585 14-C BP)')
       WRITE(*,40)
```

```
40    FORMAT(/,1X,'Enter selection:  '\)
      READ(*,*) ISET
      IF(ISET.GT.2) THEN
         NAME = 'MARINE.14C'
      ELSEIF (ISET.GT.1) THEN
         NAME = 'ATM20.14C '
      ELSE
         NAME = 'ATM10.14C '
      ENDIF
C
C  Read calibration file
C
      WRITE(*,50)
 50   FORMAT(//,3X,'READING CALIBRATION FILE---PLEASE WAIT ',/)
      OPEN(100,FILE=NAME)
      I = 1
100   IF(ISET.LT.3) THEN
         READ(100,110,END=180,ERR=970) YEAR,AGE,SIGAGE
110      FORMAT(5X,F9.1,5X,I5,5X,I2)
         X(I) = YEAR
         Y(I) = FLOAT(AGE)
         S(I) = FLOAT(SIGAGE)
      ELSE
         READ(100,120,END=180,ERR=970) YEAR,AGE
120      FORMAT(5X,F9.1,5X,I5)
         X(I) = YEAR
         Y(I) = FLOAT(AGE)
      ENDIF
      NPTS = I
      I = I + 1
      GOTO 100
180   NSAM = 0
      DO 185 J=1,4
        MREF(J) = ' '
185   CONTINUE
      YMAX = -1E30
      DO 190 J=1,NPTS
         YMAX = AMAX1(Y(J),YMAX)
190   CONTINUE
200   DO 250 I=1,LEND
         LO=LU(I)
         WRITE(LO,205) NAME
205      FORMAT(1X,'Calibration file:  ',A10,//)
         WRITE(LO,210)
210      FORMAT(' Lab #',11X,'Radiocarbon',4X,'calibrated age(s)',
     &         9X,'References')
         WRITE(LO,220)
220      FORMAT(' ',18X,'Age BP')
250   CONTINUE
300   WRITE(*,*)
      WRITE(200,'(//)')
305   WRITE(*,310)
310   FORMAT('  Enter sample ID   (or XX to end)' )
      READ(*,320) IDSAM
320   FORMAT(A12)
      IF((ID.EQ.'XX').OR.(ID.EQ.'xx')) GOTO 1000
      WRITE(*,330) COMMA
330   FORMAT('  Enter radiocarbon age BP',A,' standard error ')
      READ(*,*) RAGE,SAMSIG
      IF((RAGE.LE.0.0).OR.(RAGE.GE.YMAX)) THEN
         WRITE(*,340) 0,INT(YMAX-.5)
340   FORMAT(' VALID RADIOCARBON AGES FOR THIS DATA MUST BE BETWEEN',
     &I2,' AND',I5,' YRS BP')
         GOTO 305
      ENDIF
C
C  For Marine samples, enter reservoir correction Delta R.
C  Default Reservoir correction is 400 yrs,  Delta R = 0.
      IF(ISET.GT.2) THEN
         DELTAR=0.0
         UNCR=0.0
         WRITE(*,350)
350      FORMAT(1X,'Enter reservior correction Delta R')
         READ(*,*) DELTAR
         WRITE(*,360)
360      FORMAT(1X,'Enter Delta R standard deviation')
         READ(*,*) UNCR
      ENDIF
C
C Print sample ID and age
      DO 375 IWRITE=1,LEND
         LO=LU(IWRITE)
         WRITE(LO,370) IDSAM,RAGE,CHPM,SAMSIG
370      FORMAT(/,1X,A12,2X,F6.1,A3,F5.1,3X,\)
375   CONTINUE
C
C Subtract reservoir correction and add estimated extension to marine
C model
      IF(ISET.GT.2) THEN
         RAGE = RAGE - DELTAR
         NPTS = NPTS + 1
         X(NPTS) = 1954.
         Y(NPTS) = 493.
      ENDIF
C
C Add estimated bomb carbon influence to dataset
      NPTS = NPTS + 1
      X(NPTS) = 1955.
      Y(NPTS) = 0.
      S(NPTS) = 32.
C
C Find intercepts of Radiocarbon age with calibration curve
380   CALL INRCP(RAGE,NPTS,INTPT,NINP,INTX)
C
C Consolidate intercepts that round to the same year
C
400   DO 420 K=1,NINP-1
         INK1 = NINT(INTPT(K))
         INK2 = NINT(INTPT(K+1))
         IF(IABS(INK1-INK2).LT.1) THEN
            DO 410 K2=K+1,NINP-1
              INTPT(K2) = INTPT(K2+1)
410         CONTINUE
            NINP = NINP-1
         ENDIF
420   CONTINUE
C
C  Write to plot file PLTFIL.14C
C
      WRITE(FMT,430) NINP
430   FORMAT('(1X,A12,I2,'I2,'(F10.1,2X))')
      WRITE(300,FMT) IDSAM,NINP,(INTPT(K),K=1,NINP)
```

```
                  WRITE(LO,'(2X,A21)') SREF(KREF)
                  KREF=KREF+1
               ELSE
                  WRITE(LO,*)
               ENDIF
            ENDIF
540      CONTINUE
550      IF(KREF.LE.JREF) THEN
            WRITE(LO,'(57X,A21)') SREF(KREF)
            KREF=KREF+1
            GOTO 550
         ENDIF
590   CONTINUE
      JAD = IAD(1)
      IF(ISET.LT.3) THEN
C
C  Take calibration curve sigma to be the average of the nearest
C  points to the intercepts
C    Note: Y(INTX(I,1)) <= INTPT(I) < Y(INTX(I,2))
C      Where INTX(I,1) and INTX(I,2) are array elements of the data
C
         SIG1 = 0.0
         DO 610 I=1,NINP
           SIG1 = SIG1 + (S(INTX(I,1)) + S(INTX(I,2)))/2.0
610      CONTINUE
         SIG1 = SIG1/NINP
      ELSE
C
C For marine samples the standard deviation in Delta R takes the
C place of the unknown model calibration curve sigma.
C
         SIG1 = UNCR
      ENDIF
      DO 625 IWRITE=1,LEND
         LO=LU(IWRITE)
         WRITE(LO,620) IAD(1),IAD(2)
620   FORMAT(' Sigma**     and cal ',A2,'(cal ',A2,')  ranges:')
625   CONTINUE
C
C Find intercepts with RAGE +- SIGMA1 for 1 and 2*SAMSIG
C
      DO 890 IR=1,2
         SIGMA1 = SQRT(SAMSIG**2 + SIG1**2)
         V=RAGE + SIGMA1
         CALL INRCP(V,NPTS,EINT,NENT,NINTX)
630      V=RAGE - SIGMA1
         CALL INRCP(V,NPTS,MINT,MENT,MINTX)
C
C  Put both sets of intercepts in temporary storage.
C
C  If there are no intercepts (NENT = 0) at the old end of
C  the curve, assign the last year in the dataset
C
         IF(NENT.LE.0) THEN
             TEMPR(1) = X(1)
             INTX(1,1) = 1
             INTX(1,2) = 1
             NENT = 1
         ELSE
             DO 640 I=1,NENT
               TEMPR(I) = EINT(I)
               INTX(I,1) = NINTX(I,1)
               INTX(I,2) = NINTX(I,2)
640          CONTINUE
         ENDIF
C
C  The modern end of the curve will always have intercepts, though
C  perhaps only with bomb 14C (1955*).
C
         DO 647 I=1,MENT
            TEMPR(NENT+I) = MINT(I)
            INTX(NENT+I,1) = MINTX(I,1)
            INTX(NENT+I,2) = MINTX(I,2)
647      CONTINUE
C
C  LENT = total # of intercepts of age + sigma and age - sigma with
C  the calibration curve.
C
         LENT = MENT + NENT
C
C
C  Rank intercepts from oldest to youngest.
C
         DO 648 I=1,LENT
            RANK(I) = I
648      CONTINUE
         DO 660 I=2,LENT
            IPNT = RANK(I)
            J = I - 1
            JPNT = RANK(J)
            IF(TEMPR(IPNT).GE.TEMPR(JPNT)) GOTO 660
            KEEP = IPNT
            RANK(I) = JPNT
            DO 654 K=J-1,1,-1
               IF(K.EQ.0) GOTO 655
               KPNT = RANK(K)
               IF(TEMPR(KPNT).GT.TEMPR(IPNT)) THEN
                  RANK(K+1) = KPNT
               ELSE
                  GOTO 655
               ENDIF
654         CONTINUE
655         RANK(K+1) = KEEP
660      CONTINUE
C
C  Find ranges from ranked intercepts
C
         ICN = 0
         RS1 = RAGE + SIGMA1
         RS2 = RAGE - SIGMA1
         SKIP = .FALSE.
         DO 690 I=1,LENT-1
            IF(.NOT.SKIP) THEN
               IP1 = I+1
               IZ = INTX(RANK(I),1)
               NEXT = INTX(RANK(IP1),1)
C
C  If age +- sigma intercept is a single point, skip it
C
               IF((Y(IZ).EQ.RS1).AND.(Y(NEXT).GT.RS1)) GOTO 690
               IF((Y(IZ).EQ.RS2).AND.(Y(NEXT).LT.RS2)) GOTO 690
C
```

```
C Assign range values from TEMPR according to RANK
C
              ICN = ICN+1
              RANGE(ICN,1) = TEMPR(RANK(I))
              RANGE(ICN,2) = TEMPR(RANK(IP1))
C
C If intercept is a turning point in the curve , share it
C with the next range,  otherwise skip to next I
C
              IF((Y(IZ).NE.RS2).AND.(Y(IZ).NE.RS1)) THEN
                 SKIP = .TRUE.
              ELSE
              ELSE
                 SKIP=.FALSE.
              ENDIF
            ELSE
               SKIP=.FALSE.
            ENDIF
690       CONTINUE
          NRANG = ICN
C
C  Consolidate ranges that overlap or have gaps < 10 years
C       NRANG = # of distinct ranges
C
729       ICN = 1
730       IF(ICN.GE.NRANG) GOTO 740
          ICNP1 = ICN + 1
C
C  Replace overlapping ranges with maximum and minimum values
C
        IF((RANGE(ICN,2).GE.RANGE(ICNP1,1)).OR.
     &     ((RANGE(ICNP1,1)-RANGE(ICN,2)).LT.10.))THEN
            RANGE(ICN,1) = AMIN1(RANGE(ICNP1,1),RANGE(ICN,1))
            RANGE(ICN,2) = AMAX1(RANGE(ICNP1,2),RANGE(ICN,2))
C
C  Move rest of ranges into empty slot
           DO 735 K2=ICNP1,NRANG-1
             K2P1 = K2 + 1
             RANGE(K2,1) = RANGE (K2P1,1)
             RANGE(K2,2) = RANGE(K2P1,2)
 735       CONTINUE
           NRANG = NRANG - 1
           GOTO 730
          ENDIF
          ICN = ICN + 1
          GOTO 730
740       WRITE(FMT,741) ICN
          WRITE(300,FMT) SIGMA1,ICN,(RANGE(I,1),
     &     RANGE(I,2),I=1,ICN)
741       FORMAT('(1X,F8.1,I2,'I2,'(F10.1,2X,F10.1))')
C
C  Print ranges
C
          DO 885 IWRITE=1,LEND
            LO = LU(IWRITE)
            WRITE(LO,750) IR,CHSIG,SIGMA1
 750        FORMAT(' ',I2,1X,A,' = ',F5.1,3X\)
            DO 880 I=1,NRANG
              CALL RWRT(LO,I)
C
C  Skip to next line after 2 ranges are written
              IF((MOD(I,2).EQ.0).AND.(I.NE.NRANG)) WRITE(LO,'(/,16X,\)')
880         CONTINUE
            WRITE(LO,*)
885       CONTINUE
          SAMSIG = 2.0*SAMSIG
890     CONTINUE
C
C Form feed after 8 samples and write headings again
C
        NSAM = NSAM + 1
        IF((LP.EQ.'Y').AND.(MOD(NSAM,6).EQ.0)) THEN
           WRITE(6,891)
891        FORMAT('1RADIOCARBON CALIBRATION PROGRAM')
           WRITE(6,205) NAME
           WRITE(6,210)
           WRITE(6,220)
        ENDIF
        GOTO 300
970     WRITE(*,975)
975     FORMAT(' ERROR IN FILE READ')
1000    WRITE(*,1001)
1001    FORMAT(1X,'CLOSING FILES')
        CLOSE(100)
C
C Form feed to leave room for references if necessary
        LSAM = 5
        IF(ISET.EQ.2) LSAM = 3
        IF((MOD(NSAM,6).GE.LSAM).AND.(LP.EQ.'Y')) THEN
           WRITE(6,1002)
1002       FORMAT('1')
        ENDIF
        DO 1200 IWRITE=1,LEND
           LO = LU(IWRITE)
           WRITE(LO,'(///)')
           WRITE(LO,1003)
1003       FORMAT(1X,'References for datasets [and intervals] used:')
           IF(ISET.GT.2) THEN
              WRITE(LO,1010) REF3(1),REF3(2)
           ELSEIF (ISET.GT.1) THEN
              J = 0
              DO 1005 K=1,IREF
                 J = J + 1
                 IF(MREF(K).EQ.'Y') THEN
                    WRITE(LO,1010) REF2(J)
                    IF(K.EQ.3) THEN
                       J = J + 1
                       WRITE(LO,1010) REF2(J)
                    ENDIF
                    IF(K.GT.3) THEN
                       DO 1004 IRF=1,8
                          WRITE(LO,1010) REFAL(IRF)
1004                   CONTINUE
                    ENDIF
                 ENDIF
1005          CONTINUE
           ELSE
              WRITE(LO,1010) REF1
           ENDIF
1010       FORMAT(1X,A63)
           WRITE(LO,1015)
1015       FORMAT(/,1X,'Comments:')
```

```
      WRITE(LO,1020)
1020  FORMAT(1X,'1955* represents influence of bomb C-14')
      WRITE(LO,1030)
1030  FORMAT(1X,'0* represents a "negative" age BP')
      IF(ISET.LT.3) THEN
         WRITE(LO,1040)  CHSQD,CHSQD
1040     FORMAT(1X,'** 1 sigma = square root of (sample sigma',A,
     &       '+ curve sigma', A,')')
         WRITE(LO,1050)  CHSQD,CHSQD
1050     FORMAT(1X,'   2 sigma = square root of [(2 sample sigma)',A,
     &       '+ curve sigma', A,']')
         WRITE(LO,1060) IABS(NINT(X(1)))
1060     FORMAT(1X,'>',I5,' BC represents end of calibration data ')
      ELSE
         WRITE(LO,1070)  CHSQD,CHSQD
1070     FORMAT(1X,'** 1 sigma = square root of (sample sigma',A,
     &       '+ uncertainty in Delta R', A,')')
         WRITE(LO,1080)  CHSQD,CHSQD
1080     FORMAT(1X,'   2 sigma = square root of [(2 sample sigma)',A,
     &       '+ Delta R sigma', A,']')
         WRITE(LO,1090) IABS(NINT(X(1)))
1090     FORMAT(1X,'>',I5,' BC represents end of calibration data ')
      ENDIF
1200  CONTINUE
2000  END
C
C
      SUBROUTINE INRCP(V,N,INTPT,NINPT,XYINT)
C
C Subroutine to find the intercepts of V with the straight line
C between two points of a dataset
C
C V = Y value for which the intercepts with the function are desired
C
C INTPT = array of intercepting points
C NINPT = # of intercepts
C N = # of data points
C XYINT = element #'s of the data array that V falls between
C
C
      COMMON X(1000),Y(1000),S(1000)
C
      REAL*4 V,INTPT(40),M,B,X3
      INTEGER*2 N,NINPT,XYINT(40,1)
C
      NINPT=0
10    DO 100 I=2,N
      X1 = X(I-1)
      X2 = X(I)
      Y1 = Y(I-1)
      Y2 = Y(I)
      IF(((V.GE.Y1).AND.(V.LT.Y2)).OR.((V.LE.Y1).AND.(V.GT.Y2))) THEN
          M = (Y2-Y1)/(X2-X1)
          B = Y1 - M*X1
          X3 = (V-B)/M
          NINPT = NINPT + 1
          INTPT(NINPT) = X3
          XYINT(NINPT,1) = I-1
          XYINT(NINPT,2) = I
      ENDIF
100   CONTINUE
      RETURN
      END
C
      SUBROUTINE ABWRT(LO)
C
      COMMON /WRINT/ABINT,BPINT,ENLNE
C
      CHARACTER MARK*1,COMMA*1
      INTEGER ABINT,BPINT,LO,NBINT
      LOGICAL ENLNE
C
      COMMA = ','
      MARK = '*'
C
      NBINT = IABS(ABINT)
      IF(ABINT.LT.1954) THEN
         IF(ENLNE) THEN
            WRITE(LO,'(1X,I4\)') NBINT
         ELSE
            WRITE(LO,'(1X,I4,A\)') NBINT,COMMA
         ENDIF
      ELSE
         NBINT = 1955
         IF(ENLNE) THEN
            WRITE(LO,'(1X,I4,A\)') NBINT,MARK
         ELSE
            WRITE(LO,'(1X,I4,2A\)') NBINT,MARK,COMMA
         ENDIF
      ENDIF
      RETURN
      END
C
      SUBROUTINE BPWRT(LO)
C
      COMMON /WRINT/ ABINT,BPINT,ENLNE
C
      CHARACTER MARK*1,COMMA*1
      INTEGER ABINT,BPINT,LO
      LOGICAL ENLNE
C
      COMMA = ','
      MARK = '*'
C
      IF(BPINT.GE.0) THEN
         IF(ENLNE) THEN
            WRITE(LO,'(1X,I4\)') BPINT
         ELSE
            WRITE(LO,'(1X,I4,A\)') BPINT,COMMA
         ENDIF
      ELSE
         BPINT = 0
         IF(ENLNE) THEN
            WRITE(LO,'(1X,I4,A\)') BPINT,MARK
         ELSE
            WRITE(LO,'(1X,I4,2A\)') BPINT,MARK,COMMA
         ENDIF
      ENDIF
      RETURN
      END
```

```
C
C
      SUBROUTINE RWRT(LO,IR)
C
      COMMON X(1000),Y(1000),S(1000)
      COMMON /WRRNG/ RANGE,SIGMA1,NRANG,NPTS,JAD
C
      REAL*4 RANGE(20,2),SIGMA1
      INTEGER NRANG,IRANGE(2),BRANG(2)
      CHARACTER IAD(3)*2,ICL*4,DASH*1,JAD*2,KAD*2,RMARK*1
C
      DATA DASH/'-'/
      DATA ICL/'cal '/
C
      RANGT = ABS(RANGE(IR,2)-RANGE(IR,1))
      ISIG = NINT(SIGMA1)
C
C Round range values to nearest ten if sigma > 100 and RANGE > 100 years
C Leave out ranges that will round to the same year (or ten years).
C
      IF((ISIG.GE.100).AND.(RANGT.GE.100.)) THEN
        IRANGE(1) = NINT(RANGE(IR,1)/10.) * 10
        IRANGE(2) = NINT(RANGE(IR,2)/10.) * 10
      ELSE
         IRANGE(1)=NINT(RANGE(IR,1))
         IRANGE(2)=NINT(RANGE(IR,2))
      ENDIF
      IF(IABS(IRANGE(1)-IRANGE(2)).GT.1) THEN
C
C  Calculate BP ranges
C
         DO 100 J=1,2
            K = 2*J - 1
            IF(RANGE(IR,J).LT.0.0) THEN
              BRANG(J) = 1949 - IRANGE(J)
              IAD(K) = 'BC'
            ELSE
              BRANG(J) = 1950 - IRANGE(J)
              IAD(K) = 'AD'
            ENDIF
100      CONTINUE
C
C Check to see if range is going to print out as zero then change
C to 1, since there is no 0 AD/BC.
C
150      DO 180 J=1,2
            IF (IRANGE(J).EQ.0) THEN
              IF(RANGE(IR,J).LT.0.0) THEN
                 BRANG(J) = 1950
                 IRANGE(J) = -1
              ELSE
                 BRANG(J) = 1949
                 IRANGE(J)= 1
              ENDIF
            ENDIF
180      CONTINUE
C
C  1954 AD is last possible year, since after 1954, the bomb C-14 signal
C  overwhelmed the natural variations; therefore any range >1954 AD prints
C  1955* and 0* BP.
C
         KAD=JAD
         DO 200 J=1,2
            K=2*J-1
            IF(RANGE(IR,J).GT.1954.) THEN
               IRANGE(J) = 1955
               RMARK = '*'
C
C  Check to see if RANGE(IR,1) and RANGE (IR,2) are either both AD
C    OR both BC and the same as the heading printed for calibrated ages.
C
            IF((IAD(1).EQ.IAD(3)).AND.(IAD(1).EQ.KAD)) THEN
               WRITE(LO,'(I4,A\)') IRANGE(J),RMARK
            ELSE
               WRITE(LO,'(A4,A2,I4,A\)') ICL,IAD(K),IRANGE(J),RMARK
               KAD=IAD(1)
            ENDIF
C
C  X(1) is the first cal year for the dataset. Any range value >= X(1)
C  prints as >X(1) and >(1949-X(1)) BP though the actual range is
C  unknown
C
         ELSEIF (RANGE(IR,J).LE.X(1))THEN
            IRANGE(J) = NINT(ABS(X(1)))
            RMARK = '>'
            IF((IAD(1).EQ.IAD(3)).AND.(IAD(1).EQ.KAD)) THEN
               WRITE(LO,'(A,I4\)') RMARK,IRANGE(J)
            ELSE
               WRITE(LO,'(A4,A2,A,I4\)') ICL,IAD(K),RMARK,IRANGE(J)
               KAD=IAD(1)
            ENDIF
         ELSE
            IRANGE(J) = IABS(IRANGE(J))
            IF((IAD(1).EQ.IAD(3)).AND.(IAD(1).EQ.KAD)) THEN
               WRITE(LO,'(I4\)') IRANGE(J)
            ELSE
               WRITE(LO,'(A4,A2,I4\)') ICL,IAD(K),IRANGE(J)
               KAD=IAD(1)
            ENDIF
         ENDIF
         ENDIF
         IF(J.LT.2) WRITE(LO,'(A\)') DASH
200      CONTINUE
C
C  Write BP ranges
C
         LSTBP=1949-NINT(X(1))
         DO 350 J=1,2
            IF(J.LT.2) THEN
               WRITE(LO,310)
310            FORMAT('('\)
            ENDIF
            IF(BRANG(J).LT.0) THEN
               BRANG(J) = 0
               RMARK = '*'
               WRITE(LO,'(I4,A\)') BRANG(J),RMARK
            ELSEIF (BRANG(J).GE.LSTBP) THEN
               BRANG(J) = LSTBP
               RMARK = '>'
               WRITE(LO,'(A,I4\)') RMARK,BRANG(J)
            ELSE
               WRITE(LO,'(I4\)') BRANG(J)
            ENDIF
```

```
            IF(J.LT.2) THEN
               WRITE(LO,'(A\)') DASH
            ELSE
               WRITE(LO,320)
320            FORMAT(')',1X\)
            ENDIF
350         CONTINUE
         ELSEIF (IRANGE(1).GE.1954) THEN
            RMARK='*'
            IRANGE(1) = 1955
            WRITE(LO,'(I4,A\)') IRANGE(1),RMARK
         ENDIF
         RETURN
         END
```

ERRATUM. Format 1070 should read:

```
1070         FORMAT(1X,'** 1 sigma = square root of (sample sigma',A,
     &         '+ Delta R sigma',A,')')
```